@强制输出
@输出名("颜色调试")
类 颜色调试 : 窗口

	@布局配置([[根布局=真,宽度=-1,高度=-1,纵向布局=真,背景颜色=-1576716]])
	变量 线性布局1 : 线性布局
	@布局配置([[父布局=线性布局1,宽度=-1,高度DP=45,左内边距DP=18,右内边距DP=18,@上外边距DP=10]])
	变量 相对布局1 : 相对布局
	@布局配置([[父布局=相对布局1,对齐方式=16,内容="传统",粗体=真,字体大小=20,字体颜色=-16777216,高度=-1,ID=1]])
	变量 文本框1 : 文本框
	@布局配置([[父布局=相对布局1,高度=-1,内容="颜色值",对齐方式=16,@位于某组件右边=1,@左外边距DP=1,纵向偏移DP=2,字体大小=16,粗体=真]])
	变量 文本框2 : 文本框
	@布局配置([[父布局=相对布局1,高度=-1,@位于布局右边=真,对齐方式=17]])
	变量 线性布局2 : 线性布局
	@布局配置([[父布局=线性布局2,图片资源="../../../素材/Png/吸管简约.png",宽度DP=21,高度DP=21]])
	变量 图片框1 : 图片框
	@布局配置([[父布局=线性布局1,宽度=-1,高度=-2,对齐方式=1,纵向布局=真,左内边距DP=20,右内边距DP=20,上内边距DP=10]])
	变量 线性布局3 : 线性布局
	@布局配置([[父布局=线性布局3,宽度=-1,高度DP=40,圆角=75,卡片阴影=0,@下外边距DP=15]])
	变量 卡片布局1 : 卡片布局
	@布局配置([[父布局=卡片布局1,宽度=-1,高度=-1,透明度=1.0,背景颜色=-328966]])
	变量 线性布局4 : 相对布局
	@布局配置([[父布局=线性布局4,宽度=-1,高度=-1,左内边距DP=13,右内边距DP=3,上内边距DP=2,下内边距DP=2,@位于某组件左边=1]])
	变量 线性布局8 : 线性布局
	@布局配置([[父布局=线性布局8,宽度=-1,高度=-1,内边距=0,背景颜色=16777215,提示文本="颜色值",单行输入=真,对齐方式=16,字体颜色=-16777216,单行显示=真,字体大小=16,粗体=真,内容="#FF000000",显示光标=假]])
	变量 输入颜色值 : 编辑框
	@布局配置([[父布局=线性布局4,宽度DP=60,高度=-1,ID=1,@位于布局左边=假,@位于布局右边=真]])
	变量 查询按钮 : 相对布局
	@布局配置([[父布局=查询按钮,圆角=75,卡片阴影=0,宽度=-1,横向偏移DP=-3,高度DP=35,@位于布局中间=真,@位于布局左边=假,@位于布局右边=真]])
	变量 卡片布局4 : 卡片布局
	@布局配置([[父布局=卡片布局4,宽度=-1,高度=-1,背景颜色=-11433996]])
	变量 线性布局7 : 线性布局
	@布局配置([[父布局=查询按钮,圆角=0,卡片阴影=0,高度DP=35,@位于布局中间=真,背景颜色=-11433996,@位于布局右边=假,宽度DP=25,@位于布局左边=真]])
	变量 卡片布局5 : 卡片布局
	@布局配置([[父布局=查询按钮,@位于布局中间=真,@位于布局右边=假,内容="查询",粗体=真,字体大小=17,@右外边距DP=5,字体颜色=-1,@位于布局左边=假,横向偏移DP=-2]])
	变量 文本框3 : 文本框
	@布局配置([[父布局=线性布局3,宽度=-1,高度DP=175,圆角=45,卡片阴影=0,背景颜色=16777215]])
	变量 卡片布局2 : 卡片布局
	@布局配置([[父布局=卡片布局2,宽度=-1,高度=-1,背景颜色=-16777216]])
	变量 当前颜色显示 : 线性布局
	@布局配置([[父布局=线性布局3,宽度=-1,高度DP=45,@上外边距DP=15,纵向布局=真,@下外边距DP=10]])
	变量 线性布局6 : 线性布局
	@布局配置([[父布局=线性布局6,字体大小=18,内容="A:FF",粗体=真]])
	变量 颜色A标签 : 文本框
	@布局配置([[父布局=线性布局6,进度=255,最大进度=255]])
	变量 拖动条1 : 拖动条
	@布局配置([[父布局=线性布局3,宽度=-1,高度DP=45,@上外边距DP=15,纵向布局=真,@下外边距DP=10]])
	变量 线性布局9 : 线性布局
	@布局配置([[父布局=线性布局9,字体大小=18,内容="R:00",粗体=真]])
	变量 颜色R标签 : 文本框
	@布局配置([[父布局=线性布局9,进度=0,最大进度=255]])
	变量 拖动条2 : 拖动条
	@布局配置([[父布局=线性布局3,宽度=-1,高度DP=45,@上外边距DP=15,纵向布局=真,@下外边距DP=10]])
	变量 线性布局10 : 线性布局
	@布局配置([[父布局=线性布局10,字体大小=18,内容="G:00",粗体=真]])
	变量 颜色G标签 : 文本框
	@布局配置([[父布局=线性布局10,进度=0,最大进度=255]])
	变量 拖动条3 : 拖动条
	@布局配置([[父布局=线性布局3,宽度=-1,高度DP=45,@上外边距DP=15,纵向布局=真,@下外边距DP=10]])
	变量 线性布局11 : 线性布局
	@布局配置([[父布局=线性布局11,字体大小=18,内容="B:00",粗体=真]])
	变量 颜色B标签 : 文本框
	@布局配置([[父布局=线性布局11,进度=0,最大进度=255]])
	变量 拖动条4 : 拖动条
	@布局配置([[父布局=线性布局3,宽度=-1,高度=-1,对齐方式=17,@上外边距DP=40]])
	变量 线性布局12 : 线性布局
	@布局配置([[父布局=线性布局12,宽度DP=200,高度DP=45,圆角=30,卡片阴影=0]])
	变量 颜色选择器按钮 : 卡片布局
	@布局配置([[父布局=颜色选择器按钮,宽度=-1,高度=-1,内容="颜色选择器",对齐方式=17,粗体=真,字体大小=17,背景颜色=-328966]])
	变量 文本框8 : 文本框

	变量 颜色值 : 文本
	变量 A1 :文本
	变量 R1 :文本
	变量 G1 :文本
	变量 B1 :文本
	变量 意图1:意图
	变量 图片取色路径:文本
	事件 颜色调试:创建完毕()
		订阅事件()
		本对象.屏幕方向 = 1
		本对象.显示标题栏 =假
		本对象.状态栏字体黑色 = 真
		本对象.状态栏颜色 = -1576716
		本对象.导航栏颜色 = -1576716
		code getWindow().setSoftInputMode(2);
		拖动条1.进度 = 255
		拖动条2.进度 = 0
		拖动条3.进度 = 0
		拖动条4.进度 = 0
	结束 事件

	事件 查询按钮:被单击()
		查询指定颜色()
	结束 事件

	事件 图片框1:被单击()
		意图1.选择图片(本对象,1)
	结束 事件

	事件 颜色调试:获得返回数据(请求码 : 整数,结果码 : 整数,数据 : 启动信息)
		图片取色路径=意图1.解析图片地址(本对象,数据)
		如果 图片取色路径!=""则
			变量 图片取色弹窗:DX基础对话框
			变量 图片取色图片布局:图片取色弹窗布局
			图片取色弹窗.样式(DX样式.Material)
			图片取色弹窗.自定义布局(图片取色图片布局)
			图片取色弹窗.标题("图片取色")
			图片取色弹窗.按钮1("确定")
			图片取色弹窗.按钮2("取消")
			图片取色弹窗.显示()
			图片取色图片布局.图片框1.图片路径 = 图片取色路径
			事件 图片取色图片布局.图片框1:被触摸(来源事件 : 触摸事件) : 逻辑型
				变量 图像:位图对象 = 位图对象.从文件路径创建位图(图片取色路径)
				变量 颜色:整数 = 图像.取像素((来源事件.取横坐标()*图像.宽度/图片取色图片布局.图片框1.宽度).到整数(),(来源事件.取纵坐标()*图像.高度/图片取色图片布局.图片框1.高度).到整数())
				图片取色图片布局.文本框1.内容 = 颜色操作.颜色值到文本(颜色,真,真)
				图片取色图片布局.卡片布局1.背景颜色 = 颜色
				返回 真
			结束 事件
			事件 图片取色弹窗:按钮1被单击() : 逻辑型
				输入颜色值.内容=图片取色图片布局.文本框1.内容
				查询指定颜色()
				返回 假
			结束 事件
		结束 如果
	结束 事件

	事件 颜色选择器按钮:被单击()
		变量 消息 : DX基础对话框
		变量 布局 : 颜色布局
		消息.样式(DX样式.Material)
		消息.自定义布局(布局)
		消息.标题("颜色选择器")
		消息.按钮1("确定")
		消息.按钮2("取消")
		消息.显示()
		事件 布局.调色盘1:调色盘被触摸(颜色 : 整数,用户触摸 : 逻辑型,来自应用 : 逻辑型)
			变量 颜色值 : 文本 = 颜色操作.颜色值到文本(颜色,真,真)
			消息.内容(颜色值)
			事件 消息:按钮1被单击() : 逻辑型
				输入颜色值.内容=颜色值
				查询指定颜色()
				返回 假
			结束 事件
		结束 事件
	结束 事件

	方法 查询指定颜色()
		当前颜色显示.背景颜色 = 颜色操作.文本到颜色值(输入颜色值.内容)
		变量 当前判断颜色值:文本=输入颜色值.内容.到大写()
		变量 A值:文本
		变量 R值:文本
		变量 G值:文本
		变量 B值:文本
		如果 当前判断颜色值[0].到文本()=="#"则
			假如 当前判断颜色值.长度
				是 9
					A值=当前判断颜色值[1].到文本()+当前判断颜色值[2].到文本()
					R值=当前判断颜色值[3].到文本()+当前判断颜色值[4].到文本()
					G值=当前判断颜色值[5].到文本()+当前判断颜色值[6].到文本()
					B值=当前判断颜色值[7].到文本()+当前判断颜色值[8].到文本()
				是 7
					A值="FF"
					R值=当前判断颜色值[1].到文本()+当前判断颜色值[2].到文本()
					G值=当前判断颜色值[3].到文本()+当前判断颜色值[4].到文本()
					B值=当前判断颜色值[5].到文本()+当前判断颜色值[6].到文本()
			结束 假如

			颜色A标签.内容 = "A:" + A值
			颜色R标签.内容 = "R:" + R值
			颜色G标签.内容 = "G:" + G值
			颜色B标签.内容 = "B:" + B值
			变量 进制转换1:进制转换
			拖动条1.进度=进制转换1.十六转十(A值)
			拖动条2.进度=进制转换1.十六转十(R值)
			拖动条3.进度=进制转换1.十六转十(G值)
			拖动条4.进度=进制转换1.十六转十(B值)
		否则
			弹出提示("格式错误")
		结束 如果
	结束 方法

	事件 当前颜色显示:被单击()
		变量 菜单1 : DX菜单
		变量 当前格式 : 文本 = "当前格式:      " + 输入颜色值.内容
		变量 整数型格式 : 文本 = "整数型格式:   " + 文本到颜色值(输入颜色值.内容).到文本()
		菜单1.菜单列表({当前格式,整数型格式,"取消"})
		菜单1.样式(DX样式.MIUI)
		菜单1.显示()
		事件 菜单1:列表被点击(索引 : 整数,内容 : 文本) : 逻辑型
			假如 索引
				是 0
					置剪切板文本(本对象,输入颜色值.内容)
					弹出提示("复制成功")
				是 1
					置剪切板文本(本对象,文本到颜色值(输入颜色值.内容).到文本())
					弹出提示("复制成功")
				是 2		
					菜单1.关闭()
			结束 假如
			返回 假
		结束 事件
	结束 事件

	事件 拖动条1:进度被改变(当前进度 : 整数,是否人为改变 : 逻辑型)
		变量 当前颜色值:文本 = 颜色值判断(当前进度,拖动条2.进度,拖动条3.进度,拖动条4.进度)
		当前颜色显示.背景颜色 = 颜色操作.文本到颜色值(当前颜色值)
		输入颜色值.内容=当前颜色值
	结束 事件

	事件 拖动条2:进度被改变(当前进度 : 整数,是否人为改变 : 逻辑型)
		变量 当前颜色值:文本 = 颜色值判断(拖动条1.进度,当前进度,拖动条3.进度,拖动条4.进度)
		当前颜色显示.背景颜色 = 颜色操作.文本到颜色值(当前颜色值)
		输入颜色值.内容=当前颜色值
	结束 事件

	事件 拖动条3:进度被改变(当前进度 : 整数,是否人为改变 : 逻辑型)
		变量 当前颜色值:文本 = 颜色值判断(拖动条1.进度,拖动条2.进度,当前进度,拖动条4.进度)
		当前颜色显示.背景颜色 = 颜色操作.文本到颜色值(当前颜色值)
		输入颜色值.内容=当前颜色值
	结束 事件

	事件 拖动条4:进度被改变(当前进度 : 整数,是否人为改变 : 逻辑型)
		变量 当前颜色值:文本 = 颜色值判断(拖动条1.进度,拖动条2.进度,拖动条3.进度,当前进度)
		当前颜色显示.背景颜色 = 颜色操作.文本到颜色值(当前颜色值)
		输入颜色值.内容=当前颜色值
	结束 事件

	方法 颜色值判断(进度A : 整数,进度R : 整数,进度G : 整数,进度B : 整数) : 文本
		变量 A : 文本
		变量 R :文本
		变量 G :文本
		变量 B :文本
		变量 ARGB : 文本

		如果 进度A < 0x10  则
			A = "0" + 进度A.到十六进制()
		否则
			A = 进度A.到十六进制()
		结束 如果

		如果 进度R < 0x10  则
			R = "0" + 进度R.到十六进制()
		否则
			R = 进度R.到十六进制()
		结束 如果

		如果 进度G < 0x10  则
			G = "0" + 进度G.到十六进制()
		否则 
			G = 进度G.到十六进制()
		结束 如果

		如果 进度B < 0x10  则
			B = "0" + 进度B.到十六进制()
		否则
			B = 进度B.到十六进制()
		结束 如果

		ARGB = "#" +( A + R + G + B ).到大写()
		颜色A标签.内容 = "A:" + A.到大写()
		颜色R标签.内容 = "R:" + R.到大写()
		颜色G标签.内容 = "G:" + G.到大写()
		颜色B标签.内容 = "B:" + B.到大写()
		返回 ARGB
	结束 方法

结束 类

类 颜色布局 : 组件容器

	@布局配置([[根布局=真,宽度=-1,高度=-1,纵向布局=真,对齐方式=17]])
	变量 线性布局1 : 线性布局
	@布局配置([[父布局=线性布局1,宽度=-1,高度=1000,背景颜色=-1]])
	变量 相对布局1 : 相对布局
	@布局配置([[父布局=相对布局1,@位于布局中间=真,设置初始颜色=-1,显示明暗拖动条=真,显示透明度拖动条=真,背景颜色=-1]])
	变量 调色盘1 : 调色盘

结束 类

类 图片取色弹窗布局:组件容器

	@布局配置([[根布局=真,宽度=-1,高度=-1,对齐方式=17,纵向布局=真]])
	变量 线性布局1 : 线性布局
	@布局配置([[父布局=线性布局1,对齐方式=17]])
	变量 线性布局2 : 线性布局
	@布局配置([[父布局=线性布局2,宽度DP=40,高度DP=40,圆角=24,卡片阴影=0,@右外边距DP=10]])
	变量 卡片布局1 : 卡片布局
	@布局配置([[父布局=线性布局2,@左外边距DP=5,内容="#FFFF289C",粗体=真]])
	变量 文本框1 : 文本框
	@布局配置([[父布局=线性布局1,宽度DP=220,@上外边距DP=15,圆角=28,卡片阴影=0]])
	变量 卡片布局2 : 卡片布局
	@布局配置([[父布局=卡片布局2,宽度DP=220,图片自适应=真]])
	变量 图片框1 : 图片框

结束 类